

# Modelling Tools Manual

REVISION HISTORY			
NUMBER	DATE	DESCRIPTION	NAME
453	2016-03-17		SB

# Contents

<b>1</b>	<b>Software tools and requirements</b>	<b>1</b>
<b>2</b>	<b>Introduction the Fortran modules</b>	<b>1</b>
<b>3</b>	<b>Module: BASE_UTILS</b>	<b>2</b>
3.1	Function: TOSTR . . . . .	2
3.1.1	Examples: . . . . .	2
3.2	Subroutines: STDOUT and STDERR . . . . .	4
3.3	Function: CLEANUP . . . . .	5
3.4	Subroutine: RANDOM_SEED_INIT . . . . .	5
<b>4</b>	<b>Module: CSV_IO</b>	<b>5</b>
4.1	Overview . . . . .	5
4.2	Subroutine: CSV_OPEN_WRITE . . . . .	8
4.3	Subroutine: CSV_CLOSE . . . . .	8
4.4	Subroutine: CSV_HEADER_WRITE . . . . .	9
4.5	Function: GET_FILE_UNIT . . . . .	9
4.6	Function: GET_FREE_FUNIT . . . . .	10
4.7	Function: CHECK_UNIT_VALID . . . . .	10
4.8	Function: CHECK_FILE_OPEN . . . . .	10
4.9	Subroutine: CSV_RECORD_APPEND . . . . .	11
4.9.1	Overview . . . . .	11
4.9.2	Examples . . . . .	11
4.10	Function: CSV_RECORD_SIZE . . . . .	12
4.11	Function: CSV_FILE_LINES_COUNT . . . . .	12
4.12	Subroutine: CSV_RECORD_WRITE . . . . .	13
4.13	Subroutine: CSV_MATRIX_WRITE . . . . .	13
4.13.1	Two-dimensional matrix . . . . .	13
4.13.2	One-dimensional arrays . . . . .	14
4.14	Derived type: csv_file . . . . .	14
4.14.1	Basic Example . . . . .	15
4.14.2	Arrays of structures . . . . .	15
<b>5</b>	<b>Module: LOGGER</b>	<b>15</b>
<b>6</b>	<b>Error trapping modules: ERRORS, ASSERT, EXCEPTIONS</b>	<b>15</b>

---

<b>7</b>	<b>IEEE Arithmetics</b>	<b>16</b>
7.1	Overview . . . . .	16
7.2	IEEE Exceptions . . . . .	16
7.3	Implementation details . . . . .	18
<b>8</b>	<b>Version control: Subversion (SVN)</b>	<b>18</b>
8.1	Overview . . . . .	18
8.2	First time setup of the working copy . . . . .	18
8.3	Standard workflow . . . . .	19
8.4	GUI Tools . . . . .	19
<b>9</b>	<b>Build system: GNU make</b>	<b>19</b>
9.1	Overview . . . . .	19
9.2	Using make . . . . .	19
9.2.1	Building and running the model . . . . .	20
9.2.2	Cleanup . . . . .	20
9.2.3	Debugging . . . . .	20
9.3	Tweaking Makefile . . . . .	21
<b>10</b>	<b>Coding style</b>	<b>21</b>
<b>11</b>	<b>Final Notes</b>	<b>21</b>
<b>12</b>	<b>Index</b>	<b>22</b>

---

## Abstract

This document describes the modelling tools used for the new model environment. It is partly autogenerated. Autogeneration is incomplete at this time as most of the model components (actual for 2016-03-17).

SVN address of this document source is:

\$HeadURL: [https://svn.uib.no/aha-fortran/branches/budaev/HEDTOOLS/BASE\\_UTILS.adoc](https://svn.uib.no/aha-fortran/branches/budaev/HEDTOOLS/BASE_UTILS.adoc) \$

SVN date: \$LastChangedDate: 2016-03-17 05:27:55 +0100 (Thu, 17 Mar 2016) \$

---

## 1 Software tools and requirements

Most tools needed for the model are already available on Linux (e.g. gfortran, make, Subversion, console, midnight commander etc. On Windows they can be installed manually from their official web sites. Below are some details on the Windows software.

### Fortran (Mandatory)

Intel Fortran compiler, a commercial software available at UiB.

Free GNU Fortran distribution along with make and other tools from the Equation solution <http://www.equation.com/servlet/-equation.cmd?fa=fortran>

### Subversion (Mandatory)

Windows GUI for Subversion is TortoiseSVN: <https://tortoisesvn.net/>. It is very helpful to have also console subversion client software (Tortoise includes console tools but does not install by default). A good one is SilkSVN <https://sliksvn.com/download/>.

### Console terminal (Highly recommended)

The Windows console (cmd) is extremely weak. Conemu <https://conemu.github.io/> is a much better alternative, especially with the Far manager, a two-panel console file manager similar to the ancient Norton commander for DOS (or Midnight commander on Linux): <http://www.farmanager.com/download.php?l=en>

### Geany (Recommended)

Lightweight IDE, Editor for code and any text files. <http://www.geany.org/> Also need plugins: <http://plugins.geany.org/>

### Code::Blocks for Fortran (Recommended)

IDE for Fortran. Works with many compilers, including Intel and GNU gfortran. <http://cbfortran.sourceforge.net/> Installation by unpacking into some directory.

## 2 Introduction the Fortran modules

These modelling tools include several separate modules:

- BASE\_UTILS
- CSV\_IO
- LOGGER
- Error trapping modules
- IEEE Arithmetics module

BASE\_UTILS contains a few utility functions. CSV\_IO is for output of numerical data into the CSV (comma separated values) format files. CSV is good because it is human-readable but can still be easily imported into spreadsheets and stats packages.

Invoking the modules requires the `use` keyword in Fortran. `use` should normally be the first statements before `implicit none`:

```
program TEST

  use BASE_UTILS  ! Invoke the modules
  use CSV_IO      ! into this program

  implicit none

  character (len=255) :: REC
  integer :: i
  real, dimension(6) :: RARR = [0.1,0.2,0.3,0.4,0.5,0.6]
  character (len=4), dimension(6) :: STARR=["a1","a2","a3","a4","a5","a6"]
```

```
.....
end program TEST
```

Building the program with these modules using the command line is normally a two-step process:

build the modules, e.g.

```
gfortran -g -c ../BASE_CSV_IO.f90 ../BASE_UTILS.f90
```

This step should only be done if the source code of the modules change, i.e. quite rarely.

build the program (e.g. TEST.f90) with these modules

```
gfortran -g -o TEST.exe TEST.f90 ../BASE_UTILS.f90 ../BASE_CSV_IO.f90
```

or for a generic F95 compiler:

```
f95 -g -c ../BASE_CSV_IO.f90 ../BASE_UTILS.f90
f95 -g -o TEST.exe TEST.f90 ../BASE_UTILS.f90 ../BASE_CSV_IO.f90
```

A static library of the modules could also be built, so the other more changeable code can be just linked with the library.



#### Note

The examples above assume that the module code is located in the upper-level directory, so `../`, also the build script or Makefile should normally care about all this automatically.

---

## 3 Module: BASE\_UTILS

This module contains a few utility functions and subroutines. So far there are two useful things here: **STDOUT**, **STDERR**, **TOSTR**, **CLEANUP**, and **RANDOM\_SEED\_INIT**.

### 3.1 Function: TOSTR

**TOSTR** converts everything to a string. Accepts any numeric or non-numeric type, including integer and real (kind 4 and 8), logical and strings. Also accepts arrays of these numeric types. Outputs just the string representation of the number. Aliases: **STR** (same as **TOSTR**), **NUMTOSTR** (accepts only numeric input parameter, not logical or string)

#### 3.1.1 Examples:

Integer:

```
STRING = TOSTR(12)
produces "12"
```

Single precision real (type 4)<sup>1</sup>

```
print *, ">>", TOSTR(3.1415926), "<<"
produces >>3.14159250<<
```

Double precision real (type 8)

---

<sup>1</sup> Note that float point calculations, especially single precision (real type 4) may introduce a rounding error

---

```
print *, ">>", TOSTR(3.1415926_8), "<<"
produces >>3.1415926000000001<<
```

TOSTR also converts logical type to the "TRUE" or "FALSE" strings and can also accept character string as input. In the latest case it just output the input.

### Optional parameters

TOSTR can also accept standard Fortran format string as the second optional **string** parameter, for example:

```
print *, ">>", TOSTR(3.1415926, "(f4.2)"), "<<"
produces >>3.14<<
```

```
print *, ">>", TOSTR(12, "(i4)"), "<<"
produces >> 12<<
```

With integers, TOSTR can also generate leading zeros, which is useful for auto-generating file names or variable names. In such cases, the number of leading zeros is determined by the second optional **integer** parameter. This integer sets the template for the leading zeros, the maximum string. The exact value is unimportant, only the number of digits is used.

For example,

```
print *, ">>", TOSTR(10, 100), "<<"
produces >>010<<
```

```
print *, ">>", TOSTR(10, 999), "<<"
also produces >>010<<
```

```
print *, "File_" // TOSTR(10, 10000) // ".txt"
produces File_00010.txt
```

### Examples of arrays

It is possible to convert numeric arrays to their string representation:

```
real, dimension(6) :: RARR = [0.1,0.2,0.3,0.4,0.5,0.6]
.....
print *, ">>", TOSTR(RARR), "<<"
produces > 0.100000001 0.200000003 0.300000012 0.400000006 0.500000000 0.600000024<<
```

Fortran format statement is also accepted for arrays:

```
real, dimension(6) :: RARR = [0.1,0.2,0.3,0.4,0.5,0.6]
.....
print *, ">>", TOSTR(RARR, "(f4.2)"), "<<"
produces >> 0.10 0.20 0.30 0.40 0.50 0.60<<
```

It is possible to use array slices and array constructors with implicit do:

```
print *, ">>", TOSTR(RARR(1:4)), "<<"
print *, ">>", TOSTR( (/ (RARR(i), i=1,4) /) ), "<<"
both produce >> 0.100000001 0.200000003 0.300000012 0.400000006<<
```

or using the newer format with square brackets:

```
print *, ">>", TOSTR( [(RARR(i), i=1,4), 200.1, 400.5] ), "<<"
produces >> 0.100000001 0.200000003 0.300000012 0.400000006 200.100006 400.500000<<
```

the same with format:



```
print *, ">>", TOSTR( [(RARR(i), i=1,4), 200.1, 400.5], "(f9.3)" ), "<<"
produces >> 0.100 0.200 0.300 0.400 200.100 400.500<<
```

The subroutine TOSTR is useful because it allows to change such confusing old-style Fortran string constructions as this

```
!print new gene pool. First make file name      !BSA 18/11/13
if (gen < 10) then
  write(gen1,2902) "gen-0000000",gen
else if (gen < 100) then
  write(gen1,2903) "gen-0000000",gen
else if (gen < 1000) then
  write(gen1,2904) "gen-000000",gen
else if (gen < 10000) then
  write(gen1,2905) "gen-00000",gen
else if (gen < 100000) then
  write(gen1,2906) "gen-0000",gen
else if (gen < 1000000) then
  write(gen1,2907) "gen-000",gen
else if (gen < 10000000) then
  write(gen1,2913) "gen-00",gen
else if (gen < 100000000) then
  write(gen1,2914) "gen-0",gen
else
  write(gen1,2915) "gen-",gen
end if

if (age < 10) then
  write(gen2,2920) "age-0000",age
else if (age < 100) then
  write(gen2,2921) "age-000",age
else if (age < 1000) then
  write(gen2,2922) "age-00",age
else if (age < 10000) then
  write(gen2,2923) "age-0",age
else
  write(gen2,2924) "age-",age
end if

write(gen3,2908) gen1,"-",gen2

if (expmt < 10) then
  write(string104,2901) "HED24-",MMDD,runtag,"-E0",expmt,"-o104-genepool-",gen3,".txt"
else
  write(string104,2910) "HED24-",MMDD,runtag,"-E",expmt,"-o104-genepool-",gen3,".txt"
end if
```

to a much shorter and clear like this:

```
!print new gene pool. First make file name      !BSA 18/11/13
string104 = "HED24-" // trim(MMDD) // trim(runtag) // "-E0" // &
            TOSTR(expmt,10) // "-o104-genepool-" // &
            "gen-" // TOSTR(gen, 10000000) // "-" // &
            "age-" // TOSTR(age, 10000) // f_exten
```

### 3.2 Subroutines: STDOUT and STDERR

These subroutines output arbitrary text to the terminal, either to the standard output and standard error. While it seems trivial (standard Fortran print \*, or write() can be used), it is still good to have a dedicated standard subroutine for all outputs as we can then easily modify the code to use Matlab/R API to work with and run models from within these environments, or use a GUI

window (the least necessary feature now, but may be useful if the environment is used for teaching in future). In such cases we will then implement a specific dedicated output function and just globally swap STDOUT with something like R\_MESSAGE\_PRINT or X\_TXTGUI\_PRINT.

**STDOUT/STDERR** accept an arbitrary number of string parameters, which just represent messages placed to the output. Each parameter is printed on a new line. Trivial indeed:)



#### Important

It is useful to have two separate subroutines for stdout and stderr as they could be easily separated (e.g. redirected to different files). Redirection could be done under Windows/Linux terminal in such a simple way:

```
model_command.exe 1>output_file_stdout 2>output_file_stderr
```

Here STDOUT is redirected to output\_file\_stdout, STDERR, to output\_file\_stderr.

### Examples

```
call STDOUT("-----", &
  ch01 // " = " // ch02 // TOSTR(inumber) // " ***", &
  ch10 // "; TEST NR= " // TOSTR(120.345), &
  "Pi equals to = " // TOSTR(realPi, "(f4.2)"), &
  "-----")
```

The above code just prints a message. Note that TOSTR function is used to append numerical values to the text output (unlike standard write where values are separated by commas).

### 3.3 Function: CLEANUP

**CLEANUP** Removes all spaces, tabs, and any control characters from the input string. It is useful to make sure there are no trailing spaces in fixed Fortran strings and no spaces in file names.

Example:

```
print *, ">>", CLEANUP("This is along string blablaba"), "<<"
produces >>Thisisalongstringblablaba<<
```

### 3.4 Subroutine: RANDOM\_SEED\_INIT

**RANDOM\_SEED\_INIT** is called without parameters and just initialises the random seed for the Fortran random number generator.

Example

```
call RANDOM_SEED_INIT
```

## 4 Module: CSV\_IO

### 4.1 Overview

This module contains subroutines and functions for outputting numerical data to the **CSV (Comma Separated Values)** format (**RFC4180, CSV format**).

The typical workflow for output in CSV file format is like this:

- **CSV\_OPEN\_WRITE** - physically open CSV file for writing;

- **CSV\_HEADER\_WRITE** - physically write optional descriptive header (header is just the first line of the CSV file);
- do — start loop (1) over records (rows of data file)  
do — start loop (2) over values within the same record  
**CSV\_RECORD\_APPEND** - produce record of data values of different types, append single values, arrays or lists, usually in loop(s)  
end do — end loop (2)  
**CSV\_RECORD\_WRITE** - physically write the current record of data to the output file.
- end do — end loop (1) — go to producing the next record;
- **CSV\_CLOSE** - physically closes the output CSV file.

Thus, subs ending with **\_WRITE** and **\_CLOSE** do physical write.

This module is most suited at this moment for CSV file *output* rather than input.

This module widely uses **optional arguments**. They may or may not be present in the function/subroutine call. If not all parameters are passed, so called *named parameters* are used. That is, the name of the parameter(s) within the function is explicitly stated when the function/subroutine is called.

For example, **GET\_FREE\_FUNIT** has its both parameters optional (**max\_funit** and **file\_status**), it can be called in the standard way as below:

```
intNextunit = GET_FREE_FUNIT(200, logicalFlag)
```

It can lack any parameter:

```
intNextunit = GET_FREE_FUNIT()
```

If the first optional parameter is absent, **GET\_FREE\_FUNIT** is called as here:

```
intNextunit = GET_FREE_FUNIT(file_status=logicalFlag)
```

If both parameters present but swapped in order, it should be

```
intNextunit = GET_FREE_FUNIT(file_status=logicalFlag, max_funit=200)
```

of course, it can also be used this way:

```
intNextunit = GET_FREE_FUNIT(max_funit=200, file_status=logicalFlag)
```



#### Important

The standard way of using subroutine parameters (without explicitly setting their names) when calling subroutine works only when their are not missing and their order remains the same as in the subroutine declaration. When a function / subroutine has many parameters and optional are interspersed with mandatory, *it is probably just safer to use named parameters anyway*.

---

Files can be referred either by unit or by name, but unit has precedence (if both a provided, unit is used). There is also a derived type **csv\_file** that can be used as a single file handle. If **csv\_file** object is defined, the file name, unit and the latest operation success status can be accessed as **%name**, **%unit**, **%status** (e.g. **some\_file%name**, **some\_file%unit**).

The physical file operation error flag, **csv\_file\_status** is of logical type. It is always an optional parameter.

Here is an example of the data saving workflow:

---

```

use CSV_IO ! invoke this module first
.....
.....
! 1. Generate file name for CSV output
csv_file_append_data_name="data_genomeNR_" // TOSTR(i) // "_" // TOSTR(j) // &
                                "_" // TOSTR(k) // ".csv"
.....
! 2. open CSV file for writing
call CSV_OPEN_WRITE (csv_file_append_data_name, csv_file_append_data_unit, &
                    csv_written_ok)
if (.not. csv_written_ok) goto 1000 ! handle possible CSV error
! 3. Write optional descriptive header for the file
call CSV_HEADER_WRITE(csv_file_name = csv_file_append_data_name, &
                    header = header_is_from_this_string, &
                    csv_file_status = csv_written_ok)
.....
.....
! 4. Generate a whole record of variable (column) names
record_csv="" ! but first, prepare empty record string
call CSV_RECORD_APPEND(record_csv,["VAR_001", ("VAR_" // TOSTR(i,100),i=2,Cdip)])
! 5. physically write this variable header record to the file
call CSV_RECORD_WRITE (record=record_csv, &
                    csv_file_name=csv_file_append_data_name,&
                    csv_file_status=csv_written_ok)
if (.not. csv_written_ok) goto 1000 ! handle possible CSV error
.....
.....
! 6. Now we can write records containing actual data values, we do this
!     in two do-cycles
CYCLE_OVER_RECORDS: do l=1, Cdip
    ! 7. Prepare an empty string for the current CSV record
    record_csv=""
    CYCLE_WITHIN_RECORD: do m=1, CNRcomp
        ....
        ! do some calculations...
        ....
        ....
        ! 8. append the next value (single number: genomeNR) to the current record
        call CSV_RECORD_APPEND ( record_csv, genomeNR(l,m) )
        ....
    end do CYCLE_WITHIN_RECORD
    ! 9. physically write the current record
    call CSV_RECORD_WRITE ( record=record_csv, &
                        csv_file_name=csv_file_append_data_name,&
                        csv_file_status=csv_written_ok )
    if (.not. csv_written_ok) goto 1000 ! handle possible CSV error
    ....
end do CYCLE_OVER_RECORDS
! 10. close the CSV file when done
call CSV_CLOSE( csv_file_name=csv_file_append_data_name, &
                csv_file_status=csv_written_ok )
if (.not. csv_written_ok) goto 1000 ! handle possible CSV error

```

Although, there is a wrapper for saving the whole chunk of the data at once. A whole array or matrix (2-dimensional table) can be exported to CSV in a single command:

```

! save the whole matrix/array d_matrix to some_file.csv
call CSV_MATRIX_WRITE(d_matrix, "some_file.csv", fstat_csv)
if (.not. fstat_csv) goto 1000

```

## 4.2 Subroutine: CSV\_OPEN\_WRITE

Open CSV file for writing. May have two forms:

(1) either get three parameters:

```
character (len=*) :: csv_file_name ! file name
integer :: csv_file_unit           ! file unit
logical :: csv_file_status         ! optional status flag, TRUE if operation
                                   ! successful
```

(2) get the (single) file handle object of the derived type csv\_file

```
type(csv_file), intent(inout) :: csv_file_handle ! file handle object
```

### Example

```
type(csv_file) :: file_occ ! declare file handle object
.....
call CSV_OPEN_WRITE(file_occ) ! use file handle object
.....
call CSV_OPEN_WRITE(file_name_data1, file_unit_data1, fstat_csv) ! old style
if (.not. fstat_csv) goto 1000
```

## 4.3 Subroutine: CSV\_CLOSE

Closes a CSV file for reading or writing. May have two forms:

(1) either get three optional parameters:

```
character (len=*) :: csv_file_name ! file name
integer :: csv_file_unit           ! file unit
logical :: csv_file_status         ! optional status flag, TRUE if operation
                                   ! successful
```



### Important

At least **file name** or **unit** should be present in the subroutine call.

---

(2) get one file handle object of the derived type csv\_file

```
type(csv_file), intent(inout) :: csv_file_handle ! file handle object
```

### Example

```
type(csv_file) :: file_occ ! declare file handle object
.....
call CSV_CLOSE(file_occ) ! use file handle object
.....
call CSV_CLOSE(csv_file_name=file_name_data1, & ! old style
               csv_file_status=fstat_csv)
if (.not. fstat_csv) goto 1000
```

---

#### 4.4 Subroutine: CSV\_HEADER\_WRITE

Writes an optional descriptive header to a CSV file. The header should normally be the first line of the file.

May have two forms:

(1) either get four parameters, only the header is mandatory, but the file must be identified by name or unit:

```
character (len=*) :: csv_file_name    ! file name
integer :: csv_file_unit              ! file unit
character (len=*) :: header          ! header string
logical :: csv_file_status            ! status flag, TRUE if operation successful
```



##### Important

At least **file name** or **unit** should be present in the subroutine call.

(2) get two parameters including the header string and the file handle object of the type `csv_file`

```
character (len=*) :: header          ! mandatory CSV file header
type(csv_file) :: csv_file_handle    ! file handle object
```

##### Example

```
call CSV_HEADER_WRITE(csv_file_name=FILE_NAME_CSV1, &
    header="Example header. Total " // TOSTR(CSV_RECORD_SIZE(record_csv)) // &
    " columns of data.", csv_file_status=fstat_csv)
if (.not. fstat_csv) goto 1000
```

Here CSV file header is generated from several components, including the `CSV_RECORD_SIZE` function to count the record size.

#### 4.5 Function: GET\_FILE\_UNIT

Returns file unit associated with an existing open file name, if no file unit is associated with this name (file is not opened), return `unit=-1` and error status

Input parameters:

```
character (len=*) :: csv_file_name    ! mandatory file name
logical :: csv_file_status            ! optional status flag, TRUE if operation
                                     ! successful
```

Output parameter (function value):

```
integer :: csv_file_unit              ! unit associated with open file name
```

##### Example

```
file_unit = GET_FILE_UNIT(file_name)
```

#### 4.6 Function: GET\_FREE\_FUNIT

Returns the next free/available Fortran file unit number. Can optionally search until a specific maximum unit number.

Input parameters, optional:

```
logical  :: file_status      ! operation success status
integer  :: max_funit        ! maximum unit to search
```

Output parameter (function value):

```
integer :: file_unit           ! the first free/available file unit
```



## Important

When optional input parameters are absent, the function uses a hardwired maximum unit number, possibly depending on the computer platform and compiler used.

### Example

```
restart_file_unit_27 = GET_FREE_FUNIT()
```

#### 4.7 Function: CHECK\_UNIT\_VALID

Checks if file unit is valid, that is within the allowed range and doesn't include standard input/output/stderr units. The unit should not necessarily be linked to any file or be an open file.

Input parameter:

```
integer :: file_unit                                ! Fortran file unit to check
```

Output parameter (function value):

```
logical :: file_status      ! gets TRUE if the unit is valid
```

### Example

```
if (.not. CHECK_UNIT_VALID(csv_file_unit)) then
    csv_file_unit=GET_FREE_FUNIT(csv_file_status, MAX_UNIT)
.....
```

In this example, we check if the user provided unit is valid, if not, get the first available one.

#### 4.8 Function: CHECK FILE OPEN

Checks if a file is currently open, can optionally determine the Fortran unit associated with an open file (returns -1 if it is not open). Input parameters can be either raw form (file name or unit) or `csv_file` object. Optional `csv_file_status` can determine if the check proceeded without error (=TRUE) there was an error when trying to access the file (=FALSE). Input parameters must be either file name or unit.

Standard (verbose) form:

[illegible]

File object form:

```
type(csv_file) :: csv_file_handle
```

Output of the function is logical type, returns TRUE if the file is currently opened, FALSE otherwise.

#### Examples:

```
if (.not. CHECK_FILE_OPEN("file_out.csv")) call OPEN_OUTPUT
...
if (CHECK_FILE_OPEN(csv_file_unit=12)) goto 100
...
file_is_open = CHECK_FILE_OPEN(csv_file_name="data_out.csv", &
                               get_csv_file_unit=fileunit, &
                               csv_file_status=error_flag)
```

```
type (csv_file) :: output_handle
...
if (CHECK_FILE_OPEN(output_handle)) then
...
```

## 4.9 Subroutine: CSV\_RECORD\_APPEND

Appends one of the possible data objects to the current CSV record. Data objects could be either a single value (integer, real with single or double precision, character string) or a one-dimensional array of the above types or still an arbitrary length list of the same data types from the above list.

### 4.9.1 Overview

The first parameter of the subroutine is always character string record:

```
character (len=*) :: record          ! character string record to append data
```

The other parameters may be of any of the following types: integer (kind=4), real(kind=4), real(kind=8), character string.



#### Important

The record keeping variable can be either fixed length string or an allocatable string. But it should fit the whole record. This might be a little bit tricky if record is allocatable as `record_string=""` allocates it to an empty string. A good tip is to use the `repeat` function in Fortran to allocate the record string to the necessary value, e.g. `record=repeat(" ", MAX_RECORD)` will produce a string consisting of `MAX_RECORD` blank characters. `record` should not necessarily be an empty string initially, it could be just a whole blank string.

### 4.9.2 Examples

Append a single string to the current record:

```
call CSV_RECORD_APPEND(record_csv, "ROW_NAMES")
```

Append a single value (any of the supported types) to the current record:

```
call CSV_RECORD_APPEND(record_csv, value)      ! some variable of supported type
call CSV_RECORD_APPEND(record_csv, 123.5_8)    ! double precision literal value
```

Append a list of values (any one of the supported types) to the current record:





Function value: an integer

```
integer :: csv_file_lines_count      ! number of lines in file, -1 if file error
```

Can actually calculate the number of lines in any text file. Does not distinguish header or variable names lines in the CSV file and does not recognize CSV format.

#### Example

```
print *, "File ", CSV_FILE_LINES_COUNT("test_file.csv", succ_flag), "lines."
```

## 4.12 Subroutine: CSV\_RECORD\_WRITE

Physically writes a complete record of data to a CSV file. A record is a single row of data in the file.

This subroutine has two forms:

(1) it can either accept three parameters:

```
character (len=*) :: csv_file_name      ! file name
integer :: csv_file_unit                ! file unit
character (len=*) :: record             ! current CSV record (mandatory)
logical :: csv_file_status              ! optional operation status, TRUE if
                                      ! success
```



#### Important

The file to write the current record can be referred either by name or unit. So one of them must be present in the subroutine call.

(2) get the CSV record and the (single) file handle object of the derived type `csv_file`

```
character (len=*) :: record              ! current CSV record (mandatory)
type(csv_file) :: csv_file_handle        ! file handle object
```

#### Example

```
call CSV_RECORD_WRITE(csv_record, file_cop)      ! write current record
call LOG_MSG("Physically wrote record " // TOSTR(a) // & ! report this in some
              " to the file " // file_cop%name // &      ! logging subroutine.
              ", write status =" // TOSTR(file_cop%status))
```

Note, that file handle object is used in the above example.

## 4.13 Subroutine: CSV\_MATRIX\_WRITE

Writes a matrix of real (kind 4 or 8), integer or string values to a CSV data file. This is a shortcut allowing to write data in a single code instruction. This subroutine works either with a two-dimensional matrix or one-dimensional array (vector). The behaviour is a little different in these cases.

### 4.13.1 Two-dimensional matrix

It gets the following parameters: (1) two-dimensional data matrix (of any supported type), (2) mandatory name of the output file; (3) optional vector of column names. If the column name vector is shorter than the "column" dimension of the data matrix, the remaining columns get "COL\_XXX" names, where XXX is the consecutive column number (so they are unique). and (4) optional logical file operation success status.

```
[any supported], dimension(:, :) :: matrix ! data object, array or 2-d matrix
character (len=*) :: csv_file_name         ! file name for output
character, dimension(:) :: colnames        ! optional array of column names
logical :: csv_file_status                 ! operation status, TRUE if success
```

### Example

```
real, dimension(1:100,1:30) :: MATRIX
character (len=8), dimension(1:10) :: NAMES = ["MEAS_001", "MEAS_002", "MEAS_003", &
    "MEAS_004", "MEAS_005", "MEAS_006", "MEAS_007", "MEAS_008", "MEAS_009", "MEAS_010"]
....
! save data with column names, the first ten names are taken from the NAMES
! string array, the remaining ones are autogenerated
call CSV_MATRIX_WRITE(matrix=MATRIX, colnames=NAMES,
    csv_file_name="data_file.csv", csv_file_status=fstat_csv)
if (.not. fstat_csv) goto 1000

! save data without column names
call CSV_MATRIX_WRITE(matrix=MATRIX, csv_file_name="data_file.csv",
    csv_file_status=fstat_csv)
if (.not. fstat_csv) goto 1000
```

### 4.13.2 One-dimensional arrays

With one-dimensional array (vector), the subroutine gets (1) the array, (2) output file name, (3) logical parameter pointing if the array is saved "vertically" (as a single column, if TRUE) or "horizontally" (as a single row, if FALSE). If the vertical parameter is absent, the default TRUE (i.e. "vertical" data output) is used. There is also an alias to this subroutine, **CSV\_ARRAY\_WRITE**.

```
[any supported], dimension(:) :: array ! data object, array
character (len=*) :: csv_file_name ! file name for output
logical :: vertical ! optional parameter defining how one-
! dimensional array is saved
logical :: csv_file_status ! operation status, TRUE if success
```

### Example

```
! Here the data will be written into a single row of values
call CSV_MATRIX_WRITE (ARRAY, "data_file.csv", .FALSE., fstat_csv)
if (.not. fstat_csv) goto 1000
```

### Tip

In the simplest cases, with only the data object and the file name, CSV\_MATRIX\_WRITE can be used with a two-dimensional matrix or one-dimensional array in the same way (it's convenient during debugging):

```
real, dimension(1:100,1:20) :: MatrixX ! Matrix, two dimensional
real, dimension(1:100) :: Array_Y ! Array, one-dimensional
.....
.....
call CSV_MATRIX_WRITE(MatrixX, "file_matrixx.csv") ! write 2-d matrix
call CSV_MATRIX_WRITE(Array_Y, "file_array_y.csv") ! write 1-d array
```

## 4.14 Derived type: csv\_file

This type is used as a unitary file handle object. It has the following structure:

```

type, public :: csv_file
  character (len=MAX_FILENAME) :: name ! The name of the file
  integer :: unit = -1                ! Fortran unit associated with the file
  logical :: status = .TRUE.          ! success flag for the latest operation
end type csv_file

```

If `csv_file` object is defined, the file name, unit and the latest operation success flag can be accessed as `%name`, `%unit`, `%status` (e.g. `some_file%name`, `some_file%unit`).

#### 4.14.1 Basic Example

```

type(csv_file) :: file_occ           ! define the file handle object
....
file_occ%name="some_name.txt"        ! set file name value
....
call CSV_OPEN_WRITE(file_occ)        ! Open file for writing
....
call CSV_CLOSE(file_occ)             ! Close file

```

#### 4.14.2 Arrays of structures

This derived type can be also used as an array. An example below shows how can this be done.

```

type(csv_file), dimension(:), allocatable :: file_ABM ! Define allocatable array
.....
allocate(file_ABM(modulators))                        ! Allocate this array
.....
! now, use the array to handle many files of the same type
do j=1, modulators
  file_ABM(j)%name = "file_no_" // TOSTR(j,10) // ".csv" ! Set file handle (j)
  call CSV_OPEN_WRITE(file_ABM(j))                      ! and use it
end do

```

#### Important



The file name is set as a standard **non-allocatable** fixed string because allocatable strings may not be supported on all compiler types and versions. Notably, older GNU gfortran (prior to v.5) does not allow allocatable strings in derived types. Currently, `MAX_FILENAME=255` (can be changed in the code). There is one consequence of using fixed strings: you may have to use the Fortran `trim()` function to cut off trailing blanks if strings are concatenated. E.g. do `file_name=trim(String1) //trim(String2)` instead of `file_name=String1 //String2` or use `file_name=CLEANUP(String1 //String2)` to remove all blank and control characters.

## 5 Module: LOGGER

This module controls to log different messages during the execution of the program. The format of the messages is configurable during the run time.

To be done

## 6 Error trapping modules: ERRORS, ASSERT, EXCEPTIONS

These modules can be used for error trapping and handling.

To be done

## 7 IEEE Arithmetics

### 7.1 Overview

The model can now use the IEEE arithmetic modules. They allow exact control of the CPU math features and exceptions caused by invalid calculations, such as division by zero, overflow, underflow etc. A potential issue is that they have an optional status in the Fortran standard, so compilers do not have to implement them, although many do.



#### Important

IEEE arithmetic and exceptions are fully described in chapter 14 of this book: Adams, et al., 2009 *The Fortran 2003 Handbook*. Springer.

For example, Intel Fortran implements intrinsic IEEE arithmetics modules. GNU Fortran does not implement them until version 5.<sup>2</sup> However, there are external (non-intrinsic) IEEE modules for gfortran on the x86 (support both 32 and 64 bit) that are included into the **HEDTOOLS** bundle.



#### Important

the **fimm HPC cluster**, where calculations are normally performed, has GNU Fortran 4.8.1 and will require non-intrinsic IEEE modules. It also has the Intel Fortran which has built-in (intrinsic) IEEE modules though.

### 7.2 IEEE Exceptions

There are several exception conditions:

- IEEE\_DIVIDE\_BY\_ZERO
- IEEE\_INEXACT
- IEEE\_INVALID
- IEEE\_OVERFLOW
- IEEE\_UNDERFLOW
- IEEE\_USUAL (An array of three exceptions IEEE\_OVERFLOW, IEEE\_DIVIDE\_BY\_ZERO, IEEE\_INVALID)
- IEEE\_ALL (An array of five exceptions IEEE\_OVERFLOW, IEEE\_DIVIDE\_BY\_ZERO, IEEE\_INVALID, IEEE\_UNDERFLOW, IEEE\_INEXACT)

Normally, if the program encounters invalid arithmetic calculations, then it should crash or at least report the problem. Otherwise, correctness of calculations is not guaranteed. By default, many compilers just **ignore** invalid calculations (even many cases of division by zero, NaN<sup>3</sup> generation etc.).

In most cases NaNs and other invalid arithmetics strongly point to a bug. It is therefore wise to turn halting ON by default.

Turning arithmetic exception halting ON during the compile time requires specific compiler options.

Compiler	option	example
GNU GCC	-ffpe-trap	-ffpe-trap=zero,invalid,overflow,underflow
Intel Fortran	-fpe (/fpe)	-fpe0 (/fpe:0 on Windows)
Solaris Studio	--ftrap	--ftrap=invalid,overflow,division

<sup>2</sup> It was because GNU compiler collection is made for portability and supports many different processor architectures in addition to the most common x86 and implementation of IEEE modules is highly dependent on the CPU type and features.

<sup>3</sup> "Not a Number," a wrong arithmetic value that is not equal to itself, can result from many math errors

The IEEE module `IEEE_EXCEPTIONS` allows to control halting during the run time. For example, it is cool to switch halting ON in specific troublesome parts of the code that can normally result in invalid calculations (division by zero, invalid, inexact etc.) and control each such occurrence specifically (e.g. provide a subroutine handling and fixing the calculations).

Halting the program that encounters specific condition is controlled via `IEEE_GET_HALTING_MODE` subroutine (returns logical parameter `IEEE_DEF_MODE`). For example, for `IEEE_INVALID` it is:

```
call IEEE_GET_HALTING_MODE(IEEE_INVALID, IEEE_DEF_MODE)
```

It is also possible to set specific halting mode for specific condition. For example, to set halting ON (execution termination) on invalid arithmetic do this:

```
call IEEE_SET_HALTING_MODE(IEEE_INVALID, .TRUE.) ! Will halt on IEEE_INVALID
```

Here is an example:

```
...
! Invoke IEEE Arithmetics:
! use, non_intrinsic :: IEEE_EXCEPTIONS ! if gfortran v<5

! We normally use included auto-generated wrapper for the module
include "IEEE_wrap.inc"

IMPLICIT NONE

REAL    r, c, C0, Ap, Vc, Ke, Eb
REAL    FR1, FR2, F1, FDER

....

logical :: IEEE_MATH_FLAG, IEEE_DEF_MODE ! values for IEEE math modules

call IEEE_GET_HALTING_MODE(IEEE_INVALID, IEEE_DEF_MODE) ! Get default halting
call IEEE_SET_HALTING_MODE(IEEE_INVALID, .FALSE.)      ! NO halting from here!

...

FR2=LOG(ABS(C0)*Ap*Vc)
FR1=LOG(( (Ke+Eb)/Eb)*r*r*EXP(c*r))
F1 = FR1-FR2
FDER = c + 2./r

call IEEE_GET_FLAG(IEEE_INVALID, IEEE_MATH_FLAG) ! Get the error flag
if(IEEE_MATH_FLAG) then
  ! if IEEE exception is signalled, we cannot rely on the calculations
  ! Report the error: remember there is no halting now, the program won't stop
  write(10,*) "IEEE exception in DERIV ", r, F1, FDER, c, C0, Ap, Vc, Ke, Eb
  ! We also have to fix the calculations, e.g. equate some values to zero
  r=0.; F1=0.; FDER=0.
  call IEEE_SET_FLAG(IEEE_INVALID, .FALSE.) ! Set the error flag back to FALSE
end if

...

call IEEE_SET_HALTING_MODE(IEEE_INVALID, IEEE_DEF_MODE) ! Set default halting

END SUBROUTINE DERIV
```

### 7.3 Implementation details

We use an automatic build system (see below) which normally keeps track of the compiler and its version and IEEE modules support, there is no need to include `use, intrinsic (or non_intrinsic) :: IEEE_EXCEPTIONS` and tweak it manually depending on the compiler support. The build system automatically generates the correct include file `IEEE_wrap.inc` which should be inserted into the code in place of `use ...` statement:

```
SUBROUTINE DERIV(r,F1,FDER,c,C0,Ap,Vc,Ke,Eb)
!Derivation of equation for visual range of a predator

! Invoke IEEE Arithmetics:
! use, non_intrinsic :: IEEE_EXCEPTIONS ! if gfortran v<5

! We normally use included auto-generated wrapper for the module
include "IEEE_wrap.inc"

REAL      r,c,C0,Ap,Vc,Ke,Eb
....
```

## 8 Version control: Subversion (SVN)

AHA Repository: <https://svn.uib.no/aha-fortran>

### 8.1 Overview

Use version control not only for just managing versions, but also for organising your coding. For example, it would be good to commit changes to the server in pieces involving specific functions or parts of the model that are ready. Use the log messages to describe briefly what has been done.

For example, imagine you have implemented a new sigmoid function. Then, when it is ready, commit your change to the server with a log message like "New sigmoid function". And only after this go to the next piece of code. Then the versions you have will not be haphazard but organised into meaningful pieces. If you did several pieces in different files, e.g. sigmoid function in `Hed18.f90` and a new Makefile for building the code, do two commits:

```
svn commit Hed18.f90 -m "New sigmoid function"
...
svn commit Makefile -m "Tweaked makefile, added PGI compiler build"
```

### 8.2 First time setup of the working copy

First time setup of the working copy of the model (working directory):

- For a new project (run/experiment etc.), get into the working directory where the model code will reside (`cd`) (possibly make a new directory `mkdir`), and **checkout**: get the model code (one branch, no need to get everything!) from the server with `svn checkout https://path\_to\_branch` For example,

```
svn checkout https://svn.uib.no/aha-fortran/branches/budaev/HED18
```

This will get the HED18 into the directory HED18 within the current working directory. If we use HEDTOOLS, it should also be placed here:

```
svn checkout https://svn.uib.no/aha-fortran/branches/budaev/HED18
...
svn checkout https://svn.uib.no/aha-fortran/branches/budaev/HEDTOOLS
```

So, we now get HED18 and HEDTOOLS in our working directory.

### 8.3 Standard workflow

Now you can work within this directory. This is the standard workflow.

- **update** code from the server: `svn up`
- edit the code using any favoured tools, build, run model etc...
- **commit**, when ready (e.g. when a new has been implemented): `svn commit`

`commit` will ask you to provide a short descriptive log message. It will run the standard text editor for this by default (can be configured). But you can provide such a message just on the command line with the `-m` option:

```
svn commit Hed18.f90 -m "New sigmoid function"
```

Both `update` and `commit` can be done for the working directory as well as for specific file. E.g. to commit only the model code `Hed18.f90` do:

```
svn commit Hed18.f90
```

Both `update` and `commit` can be performed within any subdirectory of the working copy. In such cases they are limited to this subdirectory only.

### 8.4 GUI Tools

Using the GUI tools like TortoiseSVN is similar. With GUI you should just select the appropriate item from the menu list.

## 9 Build system: GNU make

### 9.1 Overview

The model currently uses a build system based on **GNU make** (Makefile). GNU make is an automated system for building source code (in fact, any digital project that requires keeping track of dependencies between multiple components.)

The make program is intended to automate the mundane aspects of transforming source code into an executable. The advantages of make over scripts is that you can specify the relationships between the elements of your program to make, and it knows through these relationships and timestamps exactly what steps need to be redone to produce the desired program each time. Using this information, make can also optimize the build process avoiding unnecessary steps.

— Mecklenburg R. *Managing Projects with GNU Make*

All the build rules for building the model executable are collected in the Makefile. If the model requires external components (e.g. non-intrinsic IEEE math modules), they will be automatically inserted.

GNU make is good because it works on diverse combinations of platforms and OSs (e.g. Linux and Windows). Some proprietary Unix platforms could supply the vendor's `make` utility that may not be compatible with the GNU `make` (e.g. Oracle Solaris includes its own `make` clone). There might be an option turning on GNU compatibility. But it is better to use the GNU `make` (`gmake` on Solaris) anyway.

### 9.2 Using make

Most basic things with the standard `Makefile` are simple.



### 9.2.1 Building and running the model

- Get a short help on the options: `make help`
- Build the model executable using default compiler: `make`
- Force rebuild the model executable with Intel compiler: `make intel`
- Force rebuild the model executable with GNU compiler: `make gnu`
- Run the current model: `make run` (on the fimm HPC cluster, this will automatically start a new batch job)

### 9.2.2 Cleanup

There are also a few options for deleting the files and data generated by the build process.

- Remove all the data generated by the model `make cleandata`
- Remove all the data files generated by the model run as well as the model executable: `make clean`
- Remove everything generated by the build system and all the data, retain the default state: `make distclean`

### 9.2.3 Debugging

The environment variable `DEBUG` controls whether the build system produces the debug symbols (`-g`) or, if NOT defined, speed-optimised machine code (`-O3`, automatic loop parallelization etc.). To build with debug support just define `DEBUG` in the manner standard for the platform/OS. For example, on Linux use:

```
$ DEBUG=1 make
```

or (`DEBUG` is now persistent)

```
$ export DEBUG=1
$ make
```

on Windows:

```
O:\WORK\MODEL\HED18>set DEBUG=1
O:\WORK\MODEL\HED18>make
```

or use `DEBUG` as a parameter to `make`, this works on all platforms:

```
$ make intel DEBUG=1
```

The `make` system keeps track of all the code components. For example, if only one has been changed, it will recompile only this. It also keeps track of whether IEEE math modules are really necessary and if the intrinsic or non-intrinsic modules are used.

For example, you may have built the model executable (`make`) and then edited the code of a module a little. Then just issue command to run batch (`make run`) on fimm. The `make` system will then automatically determine that the model executable is now out of date and recompile the changed module and build an updated executable, and only after this will start the batch job.

Another example: you just checked-out or updated (e.g. `svn up`) the model source that is tested and known to be bug-free on the fimm cluster. Now you should compile components of the program, (e.g. tweak IEEE math modules), build the executable, and finally start the executable in the cluster's batch job system. All this is done using a single command: `make run`.

```
$ svn update
$ ... some output...
$ make run
```

The system should work the same way on Windows and Linux, including the fimm HPC cluster. By editing the `Makefile` provided, one can easily tweak the behaviour of the build process, e.g. add other modules, change names, compilation options and details etc.

Microsoft Studio, Oracle Solaris Studio and other similar IDEs actually provide their own `make` systems (e.g. `nmake`, `make` or `dmake`) that work behind the scenes even if the IDE GUI is used.

## 9.3 Tweaking Makefile

To be done



### Important

A good manual on the GNU Make is this book: [Mecklenburg, R, 2005, \*Managing Projects with GNU Make\*, Third edition. O'Reilly](#). There is also the official [GNU Make Manual](#).

---

## 10 Coding style

To be done

## 11 Final Notes

There are a few other modules. I will write similar documentation for them too... Hope it is soon. There is still much to to.

The manual is generated with [AsciiDoc](#) markup processor. Later, an auto-generation of docs from the model code is planned (not first priority though).

---

## 12 Index

### A

AHA repository, [18](#)  
allocatable string, [11](#)  
    portability  
        compiler limitation, [15](#)  
array  
    one dimensional, [14](#)  
        write horizontal, [14](#)  
        write vertical, [14](#)  
    two dimensional, [13](#)  
array constructor, [3, 12](#)  
    portability  
        compiler limitation, [12](#)  
array of derived type, [15](#)  
array slice, [3, 12](#)

### B

BASE\_UTILS, [2](#)

### C

CHECK\_FILE\_OPEN, [10](#)  
CHECK\_UNIT\_VALID, [10](#)  
checkout, [18](#)  
CLEANUP, [5](#)  
column names, [12, 13](#)  
commit, [19](#)  
compiler  
    exception trapping, [17](#)  
    implementation, [18](#)  
    GNU  
        gfortran, [2, 12, 15, 16](#)  
    Intel Fortran, [16](#)  
    limitation, [12, 15, 16](#)  
compiler limitation, [12, 15](#)  
CSV\_ARRAY\_WRITE, [14](#)  
CSV\_CLOSE, [8](#)  
csv\_file, [6–9, 13, 14](#)  
CSV\_FILE\_LINES\_COUNT, [12](#)  
CSV\_HEADER\_WRITE, [9](#)  
CSV\_IO, [5](#)  
CSV\_MATRIX\_WRITE, [13](#)  
CSV\_OPEN\_WRITE, [8](#)  
CSV\_RECORD\_APPEND, [11](#)  
CSV\_RECORD\_SIZE, [12](#)  
CSV\_RECORD\_WRITE, [13](#)

### D

DEBUG, [20](#)  
derived type, [14](#)  
    array of derived type, [15](#)

### E

exception trapping, [17](#)  
    implementation, [18](#)

exceptions, [16, 17](#)  
    implementation, [18](#)

### F

file handle  
    file handle object, [6, 8, 9, 13, 14](#)  
file handle object, [6, 8, 9, 13, 14](#)  
fimm, [16](#)

### G

GET\_FILE\_UNIT, [9](#)  
GET\_FREE\_FUNIT, [10](#)  
gfortran, [2, 12, 15, 16](#)  
GNU  
    gfortran, [2, 12, 15, 16](#)  
GUI tools  
    TortoiseSVN, [19](#)

### I

IEEE arithmetic, [16–18](#)  
    exceptions, [16, 17](#)  
        implementation, [18](#)  
IEEE\_EXCEPTIONS  
    module, [17](#)  
IEEE\_wrap.inc  
    include, [18](#)  
implementation, [18](#)  
implied cycle, [3, 12](#)  
implied do, [3, 12](#)  
include, [18](#)  
Intel Fortran, [16](#)

### L

limitation, [12, 15, 16](#)  
log message, [18](#)

### M

make, [19](#)  
Makefile, [19](#)  
    make, [19](#)  
matrix, [13](#)  
    column names, [13](#)  
    two dimensional, [13](#)  
module, [17](#)

### N

named arguments, [6](#)  
NUMTOSTR, [2](#)

### O

one dimensional, [14](#)  
    write horizontal, [14](#)  
    write vertical, [14](#)  
optional arguments, [6](#)

**P**

physical disk write, [8](#), [9](#), [13](#)

portability

    compiler limitation, [12](#), [15](#)

**R**

RANDOM\_SEED\_INIT, [5](#)

record, [13](#)

repeat, [11](#)

**S**

STDERR, [5](#)

STDOUT, [5](#)

STR, [2](#)

Subversion, [18](#)

    checkout, [18](#)

    commit, [19](#)

    GUI tools

        TortoiseSVN, [19](#)

    log message, [18](#)

    TortoiseSVN, [19](#)

    update, [19](#)

svn, [18](#)

**T**

TortoiseSVN, [19](#)

TOSTR, [2](#)

two dimensional, [13](#)

**U**

update, [19](#)

**W**

workflow, [5](#), [6](#)

write horizontal, [14](#)

write vertical, [14](#)